5 EFA in a longitudinal setting

Outline

- EFA background
- EFA variations
  - ESEM, PSEM
  - Second-order, SEFA
  - Bi-factor, DSEFA
  - Target
- EFA in an SEM setting
  - MIMIC
  - EFA/CFA on EFA/CFA
- EFA in a multiple-group setting
  - EFA alignment
- **EFA in a longitudinal setting**
  - EFA longitudinal invariance testing
  - EFA longitudinal alignment
  - EFA growth modeling
- Further topics
- EFA theory

Slide 95 returns to the Outline for the presentation. We have now come to the final setting for EFA, namely longitudinal analysis.

ESEM for Longitudinal EFA: UG Ex 5.26




```
TITLE:      this is an example of an EFA at two time
            points with factor loading invariance and
            correlated residuals across time

DATA:       FILE = ex5.26.dat;

VARIABLE:   NAMES = y1-y12;

MODEL:      ! two EFA blocks t1 and t2 with
            ! metric invariance specified by using 1:
            f1-f2 BY y1-y6 (*t1 1);
            f3-f4 BY y7-y12 (*t2 1);
            y1-y6 PWITH y7-y12;
```


Slide 96 shows the User's Guide example 5.26 where you have two timepoints. The same set of 6 indicators are measured at the two timepoints. It is assumed that we have metric invariance, that is, factor loading invariance across time.

As the MODEL statement shows, this example uses two EFA blocks labeled t1 and t2, each block having an EFA with 2 factors. The label "1" indicates metric invariance. The indicator residuals are pairwise correlated using PWITH: y1 with y7, y2 with y8, etc. This may be necessary given that they represent the same indicator at the two timepoints.

An Alternative to UG Ex 5.26 Longitudinal EFA



- Random intercepts for each indicator (between person variables). The random intercepts are correlated and can be given a factor structure that is different from the within structure
- Within factors (within person variables) with auto regressions
- Auto regressions among indicator-specific residuals (not shown)

Bengt Muthén | Exploratory Factor Analysis | 97/ 139

Slide 97 shows an alternative specification for longitudinal EFA. It emphasizes the distinction between within-person and between-person factors and is in line with models such as random-intercept cross-lagged panel data (RI-CLPM). This alternative is suitable when there are more than 2 timepoints - only 2 are shown here.

As seen in the figure, each indicator has a random intercept factor - its loadings are fixed at 1 - that influences the indicator at all timepoints, thereby creating a correlation that is constant across time for the indicator. This intercept varies across individuals and can be seen as a stable trait.

The indicators also correlate across time via the factors which vary across time. These factors are regressed on the factors at the previous timepoint.

The random intercepts can either be freely correlated as shown in the figure or given a factor specification. This then leads to the distinction of between-person and within-person factors. For example, there may be one factor on between and 2 on within.

In addition, the residuals of the indicators can have auto-regressions across time. This is left out of the figure to not clutter it.

Relation of Longitudinal EFA to RI-CLPM with Multiple Indicators



- The figure corresponds to one multidimensional process in an RI-CLPM
- A second process can be added so that cross-lagged relations between the factors can be studied
- Unlike regular RI-CLPM, EFA factors are allowed (one process can be EFA and the other CFA)

Slide 98 points out the relation to RI-CLPM when RI-CLPM has multiple indicators. The figure is the same as on the previous slide and can now be seen as one half of a two-process RI-CLPM.

In an RI-CLPM of this kind there are two processes that each has 2 factors per timepoint and each has an EFA measurement model. It is also possible that one of the two processes has a CFA measurement model.

Example: Longitudinal Exploratory Factor Analysis of Positive Affect Across 7 Week Days

- PA average per day for 7 days (Tue - Mon): N = 244, T = 7
  - Dietvost et al. (2021). Grumpy of depressed? Disentangling typically developing adolescent mood from prodromal depression using experience sampling methods. Journal of Adolescence.
- PA is the average of 6 items:
  - Low arousal: Relaxed, satisfied, confident
  - High arousal: Happy, energetic, excited
- What's the factor structure for the 6 items?
  - EFA is needed
- Separate analyses of each day has disadvantages:
  - Using less information
  - Confounding trait with state variation
  - Cattell-Molenaar-Hamaker-Steyer-Eid-Geiser (LST theory)

Slide 99 shows an example of a longitudinal study of positive affect (PA) across 7 week days. PA is measured by 6 items. While the average of the items is typically used, we will instead analyze them on the item level. There are 3 low arousal items and 3 high arousal items. To not presume that the 6 items measure a single factor, we will apply an EFA with 1-3 factors.

The longitudinal EFA has several advantages over separate factor analyses of each of the 7 days in that it uses more information and does not confound trait with state variation in line with the literature of Cattell-Molenaar-Hamaker-Steyer-Eid-Geiser (LST theory).

The input for this example is shown on the next slide.

Input for Longitudinal ESEM Factor Analysis of PA Items

```
                USEVARIABLES = relax1-excit7;
                ! 6 PA items, 3 low arousal 3 high arousal:
                ! relaxed (pala1) satisfied (pala2) confident (pala3)
                ! happy (paha1) energetic (paha2) excited (paha3)
                ! 7 time points

ANALYSIS:       ESTIMATOR = MLR;

MODEL:          ! random intercepts for the 6 items
                i1 BY relax1-relax7@1;
                i2 BY satis1-satis7@1;
                i3 BY conf1-conf7@1;
                i4 BY happy1-happy7@1;
                i5 BY energ1-energ7@1;
                i6 BY excit1-excit7@1;

                ! auto-regressions among factor indicators residuals:
                relax2^-relax7^ PON relax1^-relax6^ (ar1);
                satis2^-satis7^ PON satis1^-satis6^ (ar2);
                conf2^conf7^ PON conf1^-conf6^ (ar3);
                happy2^-happy7^ PON happy1^-happy6^ (ar4);
                energ2^-energ7^ PON energ1^-energ6^ (ar5);
                excit2^-excit7^ PON excit1^-excit6^ (ar6);
```


Slide 100 shows the input for the longitudinal EFA using ESEM.

The MODEL command first specifies 6 random intercepts with loadings fixed at 1.

Next, first-order auto-regressions among the residuals of the factor indicators are specified.

The input continues on the next slide.

Longitudinal ESEM Factor Analysis Input Cont'd

```
! 2-factor ESEM with metric (loading) invariance as in UG ex 5.26
! (factor variances at first time point are automatically fixed at 1):
f11-f12 BY relax1 satis1 conf1 happy1 energ1 excit1(*1 1);
f21-f22 BY relax2 satis2 conf2 happy2 energ2 excit2(*2 1);
f31-f32 BY relax3 satis3 conf3 happy3 energ3 excit3(*3 1);
f41-f42 BY relax4 satis4 conf4 happy4 energ4 excit4(*4 1);
f51-f52 BY relax5 satis5 conf5 happy5 energ5 excit5(*5 1);
f61-f62 BY relax6 satis6 conf6 happy6 energ6 excit6(*6 1);
f71-f72 BY relax7 satis7 conf7 happy7 energ7 excit7(*7 1);

! auto-regressions among factors to reduce the number of parameters
f21-f22 ON f11-f12; f31-f32 ON f21-f22; f41-f42 ON f31-f32;
f51-f52 ON f41-f42; f61-f62 ON f51-f52; f71-f72 ON f61-f62;
i1-i6 WITH f11-f72@0;

! scalar invariance for intercepts:
[relax1-relax7] (int1);
[satis1-satis7] (int2);
[conf1-conf7] (int3);
[happy1-happy7] (int4);
[energ1-energ7] (int5);
[excit1-excit7] (int6);
[f11-f12@0 f21-f72*];
```


Slide 101 continues the input from the previous slide.

First, a 2-factor ESEM is specified for each of the 7 timepoints. This uses 7 different EFA blocks. The label "1" indicates metric invariance across the timepoints (blocks) in the same way as for multiple-group ESEM discussed earlier.

Second, the factors are regressed on themselves at the previous timepoint. The alternative of freely correlating all factors leads to many more parameters.

Third, scalar invariance is added to the metric specification by applying equality constraints for the intercepts of the 6 indicators. This addition is not necessary for the purposes of this analysis.

Standardized Factor Loading Estimates for Metric Model

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
F11 BY				
RELAX1	**0.471**	0.052	9.072	0.000
SATIS1	**0.550**	0.061	8.999	0.000
CONF1	**0.352**	0.055	6.379	0.000
HAPPY1	**0.274**	0.046	5.984	0.000
ENERG1	0.016	0.030	0.540	0.589
EXCIT1	-0.007	0.002	-3.331	0.001
F12 BY				
RELAX1	0.001	0.017	0.078	0.938
SATIS1	-0.007	0.023	-0.312	0.755
CONF1	0.023	0.037	0.623	0.533
HAPPY1	**0.308**	0.051	6.089	0.000
ENERG1	**0.475**	0.053	8.901	0.000
EXCIT1	**0.613**	0.055	11.107	0.000
F11 WITH F12				
	0.616	0.094	6.573	0.000

- Unlike EFA, ESEM analyzes a sample covariance matrix - although the factors at t=1 have variance 1, the outcomes don't: Stand'd values correspond to EFA

Slide 102 shows the two-factor EFA loadings for the 6 PA items. The two factors correspond largely to low versus high arousal items. The Happy item deviates from this in that it loads equally on both factors. The factors correlate 0.6. The correlation is significantly different from 1, rejecting the single-factor model and therefore calls into question the use of a single average score for PA.

Further Results

- Factor indicator-specific residual AR1:
  - Significant for 3 out of the 6 items: Confident, happy, energetic
- Lag1 regressions among the two within factors:
  - AR1 significant for only high-arousal PA factor
  - Insignificant cross-lagged effects between the two factors
- Percentage variance explained by the random intercepts
- Variance decompositions, reliability:
  - Eid et al. (2017). On the definition of latent-state-trait models with autoregressive effects. European Journal of Psychological Assessment, 33, 285-295

Slide 103 shows further results.

The factor indicator-specific residual AR1s are significant for 3 out of the 6 items: Confident, happy, energetic.

Regarding lag1 regressions among the two within factors, AR1 is significant for only the high-arousal PA factor, while there are insignificant cross-lagged effects between the two factors.

Other things to look for are percentage variance explained by the random intercepts and the variance decompositions, including reliability.

A relevant reference here is Eid et al. (2017). On the definition of latent-state-trait models with autoregressive effects. European Journal of Psychological Assessment, 33, 285-295.

Adding Other Variables

- Validating the need for two factors: Is there an important difference between how the two factors relate to other variables?
  - Do they have different relations to covariates or distal outcomes?
  - Such other variables can be included in longitudinal ESEM
- The time-varying covariate tired is included in the current example
- Further results are discussed in the 2023 M3 workshop slides `https://www.statmodel.com/download/Muthen2023M3Workshop.pdf`

Slide 104 brings up validation of the two EFA factors. - Is there an important difference between how the two factors relate to other variables? - Do they have different relations to covariates or distal outcomes?

Such other variables can be included in longitudinal ESEM. The time-varying covariate tired is included in the current example. Further results are discussed in the Muthén 2023 M3 workshop slides

https://www.statmodel.com/download/Muthen2023M3Workshop.pdf

The next 2 slides discuss some more technical aspects of the longitudinal EFA.

ESEM with Metric/Scalar Invariance as Part of a Bigger Model: How Does it Work?

- The unrotated model is estimated holding the unrotated loadings equal across time, fixes factor variances at 1 for the first time point, and uses uncorrelated factors. After rotation, this identifies factor covariance at the first time point, and unrestricted factor covariance matrices at all other time points (2 factors involve 2 par's decided by rotation: TECH1 shows 2 more par's than in the results). Asparouhov & Muthén (2009)
- The unrotated factors can have an AR model
- The rotation transformation is then applied to the entire model
- The necessary restriction is that if a variable is regressed on a factor in an EFA block it has to be regressed on all factors in the EFA block - the same applies for the opposite regression
- CFA factors (like random intercepts) can be combined with EFA factors
  - If one EFA factor is correlated with the CFA factor all other EFA factors in the same block must also be correlated (random intercepts are uncorrelated with the ESEM factors)

Bengt Muthén | Exploratory Factor Analysis | 105/ 139

Slide 105 shows how longitudinal EFA works using the ESEM framework of Asparouhov & Muthén (2009).

As usual, ESEM starts with an unrotated model. The unrotated model is estimated holding the unrotated loadings equal across time, fixes factor variances at 1 for the first time point, and uses uncorrelated factors. After rotation, this identifies factor covariance at the first time point, and unrestricted factor covariance matrices at all other time points (2 factors involve 2 par's decided by rotation: TECH1 shows 2 more par's than in the results).

The unrotated factors can have an AR model. The rotation transformation is then applied to the entire model.

The necessary restriction is that if a variable is regressed on a factor in an EFA block it has to be regressed on all factors in the EFA block - the same applies for the opposite regression.

CFA factors (like random intercepts) can be combined with EFA factors.

If one EFA factor is correlated with the CFA factor all other EFA factors in the same block must also be correlated (random intercepts are uncorrelated with the ESEM factors).

Longitudinal Factor Analysis Chi-Square Testing: Sample Size, Number of Variables, Number of Time Points

- P factor indicators per time point and T time points result in:
  - P*T variables = 6*7 = 42 in our example
  - The H0 model may have more parameters than the sample size
  - The H1 model has P*T*(P*T+1)/2 = 903 parameters in our example (N = 244)
- What is the quality of the regular chi-square testing of the H0 model against the unrestricted H1 model?
- Simulations based on the estimated model suggest inflated chi-square 5% reject proportions:
  - N = 250: 0.41
  - N = 500: 0.14
  - N = 1000: 0.11
  - Parameter estimates, SEs, and coverage good even at smaller N
- Small sample sizes may not be able to handle large P*T: N > P*T is needed as a bare minimum

Slide 106 discusses chi-square testing with a focus on the number of parameters.

P factor indicators per time point and T time points result in P*T variables = 6*7 = 42 in our example.

The H0 model may have more parameters than the sample size.

The H1 model has P*T*(P*T+1)/2 = 903 parameters in our example, that is, far more than the sample size of N = 244.

This raises the question of what the quality is of the regular chi-square testing of the H0 model against the unrestricted H1 model.

Simulations based on the estimated model suggest inflated chi-square 5% reject proportions:
- N = 250: 0.41
- N = 500: 0.14
- N = 1000: 0.11

This suggests that for model testing, a sample size of at least 500 should be used for this situation. However, parameter estimates, SEs, and coverage are good even at smaller N.

5.1 EFA longitudinal invariance testing

Outline

- EFA background
- EFA variations
  - ESEM, PSEM
  - Second-order, SEFA
  - Bi-factor, DSEFA
  - Target
- EFA in an SEM setting
  - MIMIC
  - EFA/CFA on EFA/CFA
- EFA in a multiple-group setting
  - EFA alignment
- EFA in a longitudinal setting
  - **EFA longitudinal invariance testing**
  - EFA longitudinal alignment
  - EFA growth modeling
- Further topics
- EFA theory

Bengt Muthén Exploratory Factor Analysis 107/ 139

Slide 107 returns to the Outline. We will now turn to invariance testing for longitudinal EFA.

Input for Longitudinal Measurement Invariance Testing

```
            USEVARIABLES = relax1-excit7;
            ! 6 PA items, 3 low arousal, 3 high arousal:
            ! relaxed (pala1) satisfied (pala2) confident (pala3)
            ! happy (paha1) energetic (paha2) excited (paha3)
            ! 7 time points

ANALYSIS:   ESTIMATOR = MLR;
            MODEL = CONFIGURAL METRIC SCALAR;

MODEL:      ! random intercepts for all 6 items:
            i1 BY relax1-relax7@1;
            i2 BY satis1-satis7@1;
            i3 BY conf1-conf7@1;
            i4 BY happy1-happy7@1;
            i5 BY energ1-energ7@1;
            i6 BY excit1-excit7@1;
            ! auto-regressions among factor indicators residuals:
            relax2^-relax7^ PON relax1^-relax6^ (ar1);
            satis2^-satis7^ PON satis1^-satis6^ (ar2);
            conf2^conf7^ PON conf1^-conf6^ (ar3);
            happy2^-happy7^ PON happy1^-happy6^ (ar4);
            energ2^-energ7^ PON energ1^-energ6^ (ar5);
            excit2^-excit7^ PON excit1^-excit6^ (ar6);
```


Slide 108 shows the input for longitudinal measurement invariance testing with EFA. This is using the PA example presented earlier.

The ANALYSIS command uses the MODEL option to test configural, metric, and scalar invariance. The rest of the specification is as shown earlier with random intercepts, and auto-regressions among factor indicator residuals.

The input continues on the next slide.

Input for Longitudinal Measurement Invariance Testing Continued: The Use of Model t

```
            ! AR1 regressions among factors
            ! to reduce the number of parameters:
            f21-f22 ON f11-f12;
            f31-f32 ON f21-f22;
            f41-f42 ON f31-f32;
            f51-f52 ON f41-f42;
            f61-f62 ON f51-f52;
            f71-f72 ON f61-f62;

            i1-i6 with f11-f72@0;

            ! 2-factor ESEM for each time point:
MODEL t1:   f11-f12 by relax1 satis1 conf1 happy1 energ1 excit1(*1);
MODEL t2:   f21-f22 by relax2 satis2 conf2 happy2 energ2 excit2(*2);
MODEL t3:   f31-f32 by relax3 satis3 conf3 happy3 energ3 excit3(*3);
MODEL t4:   f41-f42 by relax4 satis4 conf4 happy4 energ4 excit4(*4);
MODEL t5:   f51-f52 by relax5 satis5 conf5 happy5 energ5 excit5(*5);
MODEL t6:   f61-f62 by relax6 satis6 conf6 happy6 energ6 excit6(*6);
MODEL t7:   f71-f72 by relax7 satis7 conf7 happy7 energ7 excit7(*7);
```


Slide 109 continues the input with the AR1 regressions among the factors.

The bottom part of the slide shows the new feature that can be used with invariance testing. Here, an EFA is set up by ESEM for each timepoint using the MODEL t feature. This creates 7 EFA blocks, one for each timepoint. Invariance testing for these 7 EFA blocks is carried out in combination with using the ANALYSIS option model = configural metric scalar.

Longitudinal Measurement Invariance Test Results

Model	Number of Parameters	χ^2	Degrees of Freedom	P-Value
Configural	212	1246.811	733	0.0000
Metric	164	1281.246	781	0.0000
Scalar	140	1326.299	805	0.0000
Models Compared		χ^2	Degrees of Freedom	P-Value
Metric against Configural		57.341	48	0.1673
Scalar against Configural		94.980	72	0.0362
Scalar against Metric		46.660	24	0.0037

- The scalar model is typically rejected but is needed for growth modeling
- Ways out of this dilemma include:
  - Longitudinal alignment
  - Approximate invariance using PSEM or BSEM, e.g. for only the scalar part (intercepts)

Slide 110 shows the results of the measurement invariance testing. The top part of the table shows that all 3 models are rejected by chi-square. The bottom part suggests that the metric model does not fit significantly worse than the configural model, whereas the scalar models fits borderline worse than the metric model.

The scalar model is often rejected and such rejection is an obstacle to subsequent growth modeling where means are compared over time.

There are ways out of the dilemma of scalar misfit. Alignment is a key technique that will be discussed next. It is also possible to use PSEM, or the Bayesian counterpart BSEM, for approximate scalar invariance with priors for differences in the intercepts.

Automated Invariance Testing Advantages

- Also handles categorical outcomes using WLSMV with Delta and Theta parameterizations
- Automatically uses the scaling correction factors for chi-square difference testing with MLR and uses DIFFTEST with WLSMV

Slide 111 points out advantages of the automated measurement invariance testing using MODEL = configural metric scalar in combination with MODEL t.

One advantage is the handling of categorical outcomes with the WLSMV estimator using both Delta and Theta parameterization. The other advantage is the automation of scaling corrections with chi-square difference testing for MLR and difference testing using DIFFTEST with WLSMV.

Input for Longitudinal Measurement Invariance Testing with Categorical Outcomes: WLSMV

- Three 8-category ordinal items measuring 1 factor at 7 time points

```
USEVARIABLES = bkThin1f bkThin1s bkThin2s
bkThin3s bkThin4s bkThin5s bkThin6s
harmO1f harmO1s harmO2s harmO3s harmO4s harmO5s
harmO6s
takeP1f takeP1s takeP2s takeP3s takeP4s TakeP5s takeP6s ;

CATEGORICAL = bkThin1f - takeP6s;
MISSING = ALL (999);

ANALYSIS:    ESTIMATOR = WLSMV;
             MODEL = CONFIGURAL SCALAR;
```


Slide 112 shows an example of invariance testing with categorical outcomes using the WSLMV estimator. The example has 3 categorical indicators of a factor measured at 7 timepoints. The ANALYSIS command specifies WLSMV estimation and invariance testing.

The input continues on the next slide.

Input Continued

```
MODEL:       ! As before, but optional

MODEL t1:    f1 BY bkthin1f
             harmo1f
             takeP1f ;
MODEL t2:    f2 BY bkthin1s
             harmo1s
             takeP1s ;
MODEL t3:    f3 BY bkthin2s
             harmo2s
             takeP2s ;
MODEL t4:    f4 BY bkthin3s
             harmo3s
             takeP3s ;
MODEL t5:    f5 BY bkthin4s
             harmo4s
             takeP4s ;
MODEL t6:    f6 BY bkthin5s
             harmo5s
             takeP5s;
MODEL t7:    f7 BY bkthin6s
             harmo6s
             takeP6s ;
```


Slide 113 continues the input with the MODEL command which uses the MODEL t approach for the 7 timepoints. Because there is only 1 factor, EFA and ESEM are not relevant and the specification is done in CFA style.

Different Number of Categories for Different Time Points

- Adding one more time point leads to an error:
  - *** ERROR in MODEL command MODEL T1 and MODEL T8 are not equivalent. The categorical indicators in the same position for factors across time must have the same number of categories. Problem with: BKTHIN1F and BKTHIN7S
- With ML and Bayes, this can be handled by the * approach:
  - CATEGORICAL = bkThin1f - takeP7s(*);
- WLSMV cannot handle the * approach
- ML can handle it but requires numerical integration and there are typically too many dimensions of integration due to many factors
- Bayes can handle it and is feasible, but no measurement invariance chi-square testing summary is provided

Slide 114 shows a practical complication when analyzing categorical factor indicators over time, namely having different number of response categories at different timepoints. For instance, some low or high categories may not have observations at the beginning or end of the time series.

The input on the previous slides showed an analysis for the first 7 timepoints. As shown in the first bullet, adding the 8th timepoint leads to an error in that this last timepoint doesn't have the same number of categories as the previous ones.

With ML and Bayes, this can be handled but not by WLSMV.

But ML requires numerical integration of high dimension and is therefore not possible.

Bayes can do it but does not produce invariance testing.

This may need further research.

5.2 EFA longitudinal alignment

Slide 115 returns to the Outline, showing that we are still in the longitudinal setting. The next longitudinal topic is alignment.

Longitudinal Alignment

- Section 5.3 of Asparouhov & Muthén (2024). Penalized structural equation models. Structural Equation Modeling: A Multidisciplinary Journal, 31(3), 429–454.
  - 1 factor at several timepoints using the ALIGNMENT option
- PSEM can be used with DIFF priors for factor loadings so that factor variances can be estimated for t>1
  - See the Asparouhov (2023) Mplus Workshop slides 52 - `https://www.statmodel.com/download/M3TeachingSlides.pdf`
- Automated longitudinal alignment for EFA using the ALIGNMENT option

Slide 116 turns to longitudinal alignment. Th alignment approach avoids the problem of ill-fitting metric and scalar invariance over time, relying only on configural invariance.

The slide shows 3 background sources for longitudinal alignment. The 2024 PSEM paper has a section 5.3 giving an example of alignment with 1 factor measured at several timepoints. Clearly, this does not cover the EFA context.

PSEM can also be used with DIFF priors that hold factor loadings approximately equal over time. In this way, factor variance can be estimated for all but the first timepoint. This procedure is described in the 2023 workshop slides by Asparouhov.

Finally, we have the automated approach for EFA that will be described here.

Input for Longitudinal Alignment of PA Factors

```
              USEVARIABLES = relax1-excit7;
              ! 6 PA items, 3 low arousal, 3 high arousal:
              ! relaxed (pala1) satisfied (pala2) confident (pala3)
              ! happy (paha1) energetic (paha2) excited (paha3)

ANALYSIS:     ESTIMATOR = MLR;
              ALIGNMENT = FIXED; ! this is the only change
              ! to the measurement invariance input that used
              ! MODEL = CONFIGURAL etc

MODEL:        ! random intercepts for all 6 items:
              i1 BY relax1-relax7@1;
              i2 BY satis1-satis7@1;
              i3 BY conf1-conf7@1;
              i4 BY happy1-happy7@1;
              i5 BY energ1-energ7@1;
              i6 BY excit1-excit7@1;
              ! auto-regressions among factor indicators residuals:
              relax2^-relax7^ PON relax1^-relax6^ (ar1);
              satis2^-satis7^ PON satis1^-satis6^ (ar2);
              conf2^conf7^ PON conf1^-conf6^ (ar3);
              happy2^-happy7^ PON happy1^-happy6^ (ar4);
              energ2^-energ7^ PON energ1^-energ6^ (ar5);
              excit2^-excit7^ PON excit1^-excit6^ (ar6);
```


Slides 117-118 show the input for the automated alignment approach. As the comment in the ANALYSIS command says, only 1 line needs to be added to the longitudinal invariance input showed earlier, saying ALIGNMENT = FIXED.

The input continues on the next slide.

```
               ! AR1 regressions among factors
               ! to reduce the number of parameters:
               f21-f22 ON f11-f12;
               f31-f32 ON f21-f22;
               f41-f42 ON f31-f32;
               f51-f52 ON f41-f42;
               f61-f62 ON f51-f52;
               f71-f72 ON f61-f62;

               i1-i6 with f11-f72@0;

               ! 2-factor ESEM for each time point:
MODEL t1:      f11-f12 by relax1 satis1 conf1 happy1 energ1 excit1(*1);
MODEL t2:      f21-f22 by relax2 satis2 conf2 happy2 energ2 excit2(*2);
MODEL t3:      f31-f32 by relax3 satis3 conf3 happy3 energ3 excit3(*3);
MODEL t4:      f41-f42 by relax4 satis4 conf4 happy4 energ4 excit4(*4);
MODEL t5:      f51-f52 by relax5 satis5 conf5 happy5 energ5 excit5(*5);
MODEL t6:      f61-f62 by relax6 satis6 conf6 happy6 energ6 excit6(*6);
MODEL t7:      f71-f72 by relax7 satis7 conf7 happy7 energ7 excit7(*7);
```

Longitudinal Alignment Results: Measurement Part

APPROXIMATE MEASUREMENT INVARIANCE
(NONINVARIANCE) FOR TIMES

Intercepts/Thresholds							
RELAX1	1	2	3	4	5	6	7
SATIS1	1	2	3	4	5	6	7
CONF1	1	2	3	4	5	6	7
HAPPY1	1	2	3	4	5	6	7
ENERG1	1	2	3	4	5	6	7
EXCIT1	1	2	3	4	5	6	7
Loadings for F11							
RELAX1	1	2	3	4	5	6	7
SATIS1	1	2	3	4	5	6	7
CONF1	1	2	3	4	5	6	7
HAPPY1	1	2	3	4	5	6	7
ENERG1	1	2	3	4	5	6	7
EXCIT1	1	2	3	4	5	6	7
Loadings for F12							
RELAX1	1	2	3	4	5	6	7
SATIS1	1	2	3	4	5	6	7
CONF1	1	2	3	4	5	6	7
HAPPY1	1	2	3	4	5	6	7
ENERG1	1	2	3	4	5	6	7
EXCIT1	1	2	3	4	5	6	7

- No parentheses means no measurement noninvariance: All parameters are deemed invariant (scalar invariance)

Slide 119 shows the results of the alignment for the measurement part of the model. The rows represent the measurement parameters. They are intercepts for the 6 indicators and 6 loadings for each of the two EFA factors. The columns represent the 7 timepoints. Entries with parentheses are non-invariant. Because no entry has a parenthesis, all measurement parameters are invariant as judged by alignment.

FACTOR INTERCEPT COMPARISON AT THE
5% SIGNIFICANCE LEVEL IN DESCENDING ORDER

Results for Factor F11

Factor Ranking	Time	Factor Intercept	Times With Significantly Smaller Factor Intercept
1	5	0.299	1 3 7
2	4	0.002	
3	1	0.000	7
4	2	-0.114	
5	6	-0.137	
6	3	-0.152	
7	7	-0.336	

Results for Factor F12

Factor Ranking	Time	Factor Intercept	Times With Significantly Smaller Factor Intercept
1	5	0.300	1 3 6
2	4	0.236	1 3 6
3	7	0.095	6
4	1	0.000	6
5	2	-0.046	
6	3	-0.055	
7	6	-0.302	

Slide 120 shows the results of the alignment for the factor means. The top part of the table shows the means for the F11 factor. The means are listed from largest to smallest. It is seen that time 5 has the largest mean and that it is significantly larger than the means at times 1, 3, and 7. The factor mean at time 1 is fixed at zero as a result of using the FIXED option.

5.3 EFA growth modeling

Outline

- EFA background
- EFA variations
  - ESEM, PSEM
  - Second-order, SEFA
  - Bi-factor, DSEFA
  - Target
- EFA in an SEM setting
  - MIMIC
  - EFA/CFA on EFA/CFA
- EFA in a multiple-group setting
  - EFA alignment
- EFA in a longitudinal setting
  - EFA longitudinal invariance testing
  - EFA longitudinal alignment
  - **EFA growth modeling**
- Further topics
- EFA theory

Bengt Muthén Exploratory Factor Analysis 121/ 139

Slide 121 returns to the Outline of the presentation and shows that we have now reached the final longitudinal EFA topic of growth modeling.

Input for EFA Growth Modeling: PSEM GEOMIN Priors

```
            USEVARIABLES = relax1-excit7;

ANALYSIS:   ESTIMATOR = MLR;
            STARTS = 20;
            ITERATIONS = 2000;

MODEL:      f11-f12 BY relax1*1 satis1 conf1
            happy1 energ1 excit1 (a1-a12);
            f21-f22 BY relax2*1 satis2 conf2
            happy2 energ2 excit2 (a1-a12);
            f31-f32 BY relax3*1 satis3 conf3
            happy3 energ3 excit3 (a1-a12);
            f41-f42 BY relax4*1 satis4 conf4
            happy4 energ4 excit4 (a1-a12);
            f51-f52 BY relax5*1 satis5 conf5
            happy5 energ5 excit5 (a1-a12);
            f61-f62 BY relax6*1 satis6 conf6
            happy6 energ6 excit6 (a1-a12);
            f71-f72 BY relax7*1 satis7 conf7
            happy7 energ7 excit7 (a1-a12);

            f11-f12@1;

            f11 WITH f12;
            f21 WITH f22;
            f31 WITH f32;
            f41 WITH f42;
            f51 WITH f52;
            f61 WITH f62;
            f71 WITH f72;

            i1 s1 q1 c1 | f11@0 f21@.1 f31@.2 f41@.3
            f51@.4 f61@.5 f71@.6;
            i2 s2 q2 c2 | f12@0 f22@.1 f32@.2 f42@.3
            f52@.4 f62@.5 f72@.6;

            q1@0; q2@0; c1@0; c2@0;

            ! scalar invariance for intercepts:
            [relax1-relax7] (int1);
            [satis1-satis7] (int2);
            [conf1-conf7] (int3);
            [happy1-happy7] (int4);
            [energ1-energ7] (int5);
            [excit1-excit7] (int6);

            ! time-invariant residual variances:
            relax1 satis1 conf1 happy1 energ1 excit1 (v1-v6);
            relax2 satis2 conf2 happy2 energ2 excit2 (v1-v6);
            relax3 satis3 conf3 happy3 energ3 excit3 (v1-v6);
            relax4 satis4 conf4 happy4 energ4 excit4 (v1-v6);
            relax5 satis5 conf5 happy5 energ5 excit5 (v1-v6);
            relax6 satis6 conf6 happy6 energ6 excit6 (v1-v6);
            relax7 satis7 conf7 happy7 energ7 excit7 (v1-v6);

MODEL PRIORS:   a1-a12 ~ GEOMIN(2,1,0.0001);

OUTPUT:         STANDARDIZED TECH4;

PLOT:           TYPE = PLOT3;
```


Slide 122 shows the input for EFA growth modeling of the PA example.

The left column shows the EFA measurement modeling for the 2 factors at each of the 7 timepoints. It is handled by PSEM. The factor loadings are labeled and given GEOMIN priors in the MODEL PRIORS command. Note that the labels are the same at the 7 different timepoints so that metric invariance is imposed. The GEOMIN settings are shown in the EFA Theory section of the presentation.

In the right column, the growth model is specified for each of the two EFA factors using cubic growth.

Scalar invariance is imposed by intercept equalities across time. This is necessary for the growth modeling.

For parsimony, time invariance of the residual variances is also specified.